<xsl:stylesheet version="1.0" xmlns:xsl="http://www.w3.org/1999/XSL/Transform">
	<xsl:output indent="yes" method="html"/>
        <xsl:variable name="msgCount" select="/RPCReply/@count"/>
        <xsl:variable name="page" select="/RPCReply/@page"/>
        <xsl:variable name="elements" select="/RPCReply/@elements"/>
        <xsl:template match="/*/com.transcargo.posts.Cargo-array">
            Найдено предложений по грузам: <xsl:value-of select="$msgCount"/><br/>
            <table border="0" cellspacing="1" cellpadding="0" width="100%" bgcolor="#000000">
            <tr bgcolor="#A0A0A0"><th rowspan="2">Страна</th>
                <th rowspan="2">Дата</th>
                <th colspan="2">Маршрут</th>
                <th rowspan="2">Груз</th>
                <th rowspan="2">Кузов</th>
                <th rowspan="2">Контакт</th>
                <th rowspan="2">&#160;</th>
            </tr>
            <tr bgcolor="#A0A0A0"><th>Загрузка</th><th>Разгрузка</th></tr>
            <xsl:apply-templates select="com.transcargo.posts.Cargo"/>
            </table><br/>
            <xsl:if test="$msgCount &gt; $elements">
                <xsl:choose>
                    <xsl:when test="$page=0"><b>1</b>&#160;</xsl:when>
                    <xsl:otherwise><a><xsl:attribute name="href">javascript:sfrom('<xsl:value-of select="0"/>');</xsl:attribute>1</a>&#160;</xsl:otherwise>
                </xsl:choose>
                <xsl:if test="$elements &lt; $page - ($elements * 3)">...&#160;</xsl:if>
                <xsl:variable name="prevPage" select="($msgCount - 1) - (($msgCount - 1) mod $elements) - $elements"/>
                <xsl:choose>
                    <xsl:when test="$page - $elements * 3 &gt;= $elements">
                        <xsl:choose>
                            <xsl:when test="$page + $elements * 3 &lt; $prevPage">
                                <xsl:call-template name="disp_">
                                    <xsl:with-param name="min" select="$page - $elements * 3"/>
                                    <xsl:with-param name="max" select="$page + $elements * 3"/>
                                </xsl:call-template>
                            </xsl:when>
                            <xsl:otherwise>
                                <xsl:call-template name="disp_">
                                    <xsl:with-param name="min" select="$page - $elements * 3"/>
                                    <xsl:with-param name="max" select="$prevPage"/>
                                </xsl:call-template>
                            </xsl:otherwise>
                        </xsl:choose>
                    </xsl:when>
                    <xsl:otherwise>
                        <xsl:choose>
                            <xsl:when test="$page + $elements * 3 &lt; $prevPage">
                                <xsl:call-template name="disp_">
                                    <xsl:with-param name="min" select="$elements"/>
                                    <xsl:with-param name="max" select="$page + $elements * 3"/>
                                </xsl:call-template>
                            </xsl:when>
                            <xsl:otherwise>
                                <xsl:call-template name="disp_">
                                    <xsl:with-param name="min" select="$elements"/>
                                    <xsl:with-param name="max" select="$prevPage"/>
                                </xsl:call-template>
                            </xsl:otherwise>
                        </xsl:choose>
                    </xsl:otherwise>
                </xsl:choose>
                <xsl:if test="$page + ($elements * 3) &lt; $prevPage">...&#160;</xsl:if>
                <xsl:choose>
                    <xsl:when test="$page + $elements &gt;= $msgCount">
                        <b><xsl:value-of select="floor(($msgCount - 1) div $elements) + 1"/></b>&#160;
                    </xsl:when>
                    <xsl:otherwise>
                        <a><xsl:attribute name="href">javascript:sfrom('<xsl:value-of select="($msgCount - 1) - (($msgCount - 1) mod $elements)"/>');</xsl:attribute><xsl:value-of select="floor((($msgCount - 1) div $elements) + 1)"/></a>
                    </xsl:otherwise>
                </xsl:choose>
            </xsl:if>
        </xsl:template>
        <xsl:template match="com.transcargo.posts.Cargo">
        <tr bgcolor="#ffffff">
            <td rowspan="2"><b><xsl:value-of select="country__from" /></b><br/><br/>Размещено:<br/><xsl:value-of select="date__posted" /></td>
            <td rowspan="2" ><xsl:choose>
                <xsl:when test="permanent='true'">Постоянно, <xsl:value-of select="times"/> раз(а) в неделю</xsl:when>
                <xsl:otherwise><xsl:value-of select="date__load" /> - <xsl:value-of select="date__unload" /></xsl:otherwise>
                </xsl:choose>
                <xsl:if test="urgent='true'"><br/><b>Срочно! Загрузить до <xsl:value-of select="urgent__time"/></b></xsl:if>
            </td>
            <td><xsl:value-of select="city__from"/><xsl:if test="not(region__from = '')"><br/><xsl:value-of select="region__from"/></xsl:if><xsl:if test="not(country__from = '')"><br/><xsl:value-of select="country__from"/></xsl:if></td>
            <td><xsl:value-of select="city__to"/><br/>
                <xsl:if test="not(region__to = '')"><xsl:value-of select="region__to"/><br/></xsl:if>
                <xsl:value-of select="country__to"/>
                <xsl:if test="passing='true'"><br/><b>Попутный транспорт</b></xsl:if>
            </td>
            <td><xsl:value-of select="cargo"/><br/><xsl:if test="combined='true'"><b>Сборный груз</b><br/></xsl:if><xsl:if test="customs='true'"><b>Под таможенный контроль</b><br/></xsl:if></td>
            <td>
                <xsl:value-of select="type"/>&#160;&#160;<xsl:value-of select="weight"/> т./<xsl:value-of
                    select="volume"/> м<sup>3</sup><br/>
                <xsl:variable name="cntload" select="count(loading/string)"/>
                Загрузка <xsl:for-each select="loading/string"><xsl:value-of select="."/><xsl:if test="not(position()=$cntload)">, </xsl:if></xsl:for-each>
		<xsl:if test="cars &gt; 1"><br/><b>Требуется <xsl:value-of select="cars"/> машин(ы) в неделю</b></xsl:if>
                <xsl:if test="tir='true'">[TIR] </xsl:if><xsl:if test="not(adr='0')">[Класс ADR: <xsl:value-of
                    select="adr"/>]</xsl:if>
            </td>
            <td rowspan="2"><a><xsl:attribute name="href">company.jsp?id=<xsl:value-of select="user__id"/></xsl:attribute><xsl:value-of select="org__name"/></a><br/>
		<xsl:choose>
			<xsl:when test="count(contact__name) &gt; 0">
				<a><xsl:attribute name="href">contact.jsp?id=<xsl:value-of select="user__id"/></xsl:attribute><xsl:value-of select="contact__name"/></a><br/>
				<xsl:value-of select="contact__phone1"/><br/>
				<xsl:if test="count(contact__phone2) &gt; 0"><xsl:value-of select="contact__phone2"/><br/></xsl:if>
				<xsl:if test="count(contact__phone3) &gt; 0"><xsl:value-of select="contact__phone3"/><br/></xsl:if>
				<xsl:if test="count(contact__icq) &gt; 0">ICQ: <xsl:value-of select="contact__icq"/><br/></xsl:if>
				<xsl:if test="count(contact__skype) &gt; 0"><xsl:value-of select="contact__skype"/><br/></xsl:if>
			</xsl:when>
			<xsl:otherwise><xsl:value-of select="country"/><br/><xsl:value-of select="org__phone"/></xsl:otherwise>
		</xsl:choose></td>
	    <td rowspan="2"><a><xsl:attribute name="href">cargodetails.jsp?id=<xsl:value-of select="id"/></xsl:attribute>Подробнее</a></td>
        </tr>
        <tr bgcolor="#ffffff"><td colspan="4"><xsl:choose><xsl:when test="(bid='0') or (currency='0')">Ставка договорная. </xsl:when><xsl:otherwise>Ставка <xsl:value-of
                select="bid"/>&#160;<xsl:value-of select="currency"/>. </xsl:otherwise></xsl:choose>Форма оплаты: <xsl:value-of
                select="payment__type"/>. <xsl:value-of select="comments"/></td></tr>
        <xsl:if test="not(position()=last())"><tr bgcolor="#EEEEEE"><td colspan="8">&#160;</td></tr></xsl:if>
    </xsl:template>
    <xsl:template name="disp_">
        <xsl:param name="min"/>
        <xsl:param name="max"/>
        <xsl:if test="$min &lt; $max">
            <xsl:call-template name="page_">
                <xsl:with-param name="i" select="$min"/>
                <xsl:with-param name="max" select="$max"/>
            </xsl:call-template>
        </xsl:if>
        <xsl:if test="$min = $max">
            <xsl:choose>
                <xsl:when test="$page = $min"><b><xsl:value-of select="$page"/></b>&#160;</xsl:when>
                <xsl:otherwise><a><xsl:attribute name="href">javascript:sfrom('<xsl:value-of select="$min"/>');</xsl:attribute><xsl:value-of select="floor($min div $elements) + 1"/></a>&#160;</xsl:otherwise>
            </xsl:choose>
        </xsl:if>
    </xsl:template>
    <xsl:template name="page_">
        <xsl:param name="i"/>
        <xsl:param name="max"/>
        <xsl:choose>
            <xsl:when test="$i=$page"><b><xsl:value-of select="floor($page div $elements) + 1"/></b>&#160;</xsl:when>
            <xsl:otherwise><a><xsl:attribute name="href">javascript:sfrom('<xsl:value-of select="$i"/>');</xsl:attribute><xsl:value-of select="floor($i div $elements) + 1"/></a>&#160;</xsl:otherwise>
        </xsl:choose>
        <xsl:if test="$i &lt; $max">
          <xsl:call-template name="page_">
            <xsl:with-param name="i" select="$i + $elements"/>
            <xsl:with-param name="max" select="$max"/>
          </xsl:call-template>
        </xsl:if>
    </xsl:template>
</xsl:stylesheet>
